2024.2.27
電子ルーレット
薮哲郎

 概略
　以下の仕様を持つ電子ルーレットを作成する。マイコンArduino（アルデューイノ）を使用する。

LEDは6個
点灯するLEDが変わるたびに「ピッ」とスピーカーから音が出る
プッシュボタンを1個

　まず、回路をブレッドボード上に組んで動作確認を行い、次にユニバーサル基板上に、はんだ付けをして組む。ブレッドボード上に組んだ回路を図1に示し、ユニバーサル基板上に組んだ回路を図2に示す。
[image: 電子機器の部品

中程度の精度で自動的に生成された説明]
[bookmark: _Ref160186781]図1　ブレッドボード上に組んだ回路
[image: 木の回路

中程度の精度で自動的に生成された説明] [image: 電子機器, 回路 が含まれている画像

自動的に生成された説明]
[bookmark: _Ref160126892]図2　ユニバーサル基板上に組んだ回路
　使用部品を表1に示す。表1の下側半分はユニバーサル基板上に組む場合に追加する部品である。ユニバーサル基板上に組む場合、Arduinoのボードは使用せず、Arduinoに搭載されているマイコンATmega328P単体を使用する。

表1　使用部品
	部品名
	個数

	Arduino（マイコン）
	1

	赤色LED
	6

	抵抗　1 kΩ
	7

	抵抗　10 kΩ
	1

	プッシュスイッチ
	1

	圧電ブザー
	1

	

	ATmega328P
	1

	水晶発振子　16 MHz
	1

	コンデンサ　22 pF
	2

	IC用ソケット　28 pin
	1

	ピンソケット　3 pin
	1



　ブレッドボード上に組むときの回路図を図3に示す。
[image: ]
[bookmark: _Ref159863554]図3　ブレッドボード上に組むときの回路図
Arduinoとは
　Arduinoはマイコンボードの一種である。プログラムを組むのが非常に楽なので、広く普及している。パソコン上で動作するArduino IDEというアプリでプログラムを組み、USBケーブルで接続したArduinoにプログラムを送り込む。プログラムを送り込んだ後はArduino単独で動作する。
　類似の機能を持つマイコンボードとしてはmicro:bitも有名である。
回路の解説
　2～7番端子は出力モードに設定し、LEDを駆動する。出力端子に直接LEDを接続すると、電流が流れすぎてArduinoとLEDが壊れる恐れがあるため、1 kWの電流制限用抵抗を挿入する。
　8番端子は出力モードに設定し、圧電ブザーを接続する。圧電ブザーを直接接続すると、突入電流のためArduinoの出力端子にダメージを与える恐れがあるので、1 kWの電流制限用抵抗を挿入する。
　9番端子は入力モードに設定し、スイッチのon/offを受け取る。スイッチがoffのとき9番端子は5 V、onのとき0 Vになる。
　回路中の記号 [image: ] はアース（グランド (Ground, GND)、接地と言うこともある）を表す。アースは電位の基準点であり、その場所の電位は0 Vである。電子回路において、「ある場所の電圧」は「アースの場所からの電圧」を表す。回路中に [image: ] で表された場所同士は接続される。
 電気の予備知識
電子回路における色使い
　電子回路においては「赤色」は電源、「黒色」はアースを表す。テスタで測定するとき、赤色のプローブは電源側、黒色のプローブはアース側にあてる。ブレッドボードには「赤線」と「青線」が引かれている。赤線は電源の＋端子に接続し（Arduinoの場合は5 Vと書いてある端子）、青線はアース（GNDと書いてある端子）に接続する。
ブレッドボード
　ブレッドボードは回路を試作する大変便利なツールである。図4に示す形をしている。
[image: グラフィカル ユーザー インターフェイス, アプリケーション, テーブル, Excel

自動的に生成された説明]
[bookmark: _Ref159863514]図4　ブレッドボード[footnoteRef:1] [1:  https://burariweb.info/electronic-work/electronic-work-bread-board.html より引用] 

　穴の底はつながっている。穴に回路部品やワイヤーを差し込むことで、回路を作成する。
　電源ラインのエリアは「青線」「赤線」の穴がつながっている。部品配置エリアは「黒線」の5個の穴がつながっている
[image: ダイアグラム

自動的に生成された説明]
[bookmark: _Ref159863527]図5　ブレッドボードの構造[footnoteRef:2] [2:  https://shop.sunhayato.co.jp/blogs/problem-solving/breadboard　より引用] 

　内部構造は図5の通りである。バネが部品の足やワイヤーを挟む構造となっている。接触不良がおこりやすいので、ブレッドボードは試作に最適だが、長期にわたる安定動作させるものには向かない。私の経験では1年以内であれば、ブレッドボードで組んだ回路は動くが、2～3年すると、どこかに接触不良が発生し動かなくなる。長期にわたって使い続けるものは、ユニバーサル基板上に、はんだ付けをして作ることを推奨する。
部品
	名称
	回路記号
	外観
	注意点

	抵抗
	[image: ]
	[image: ]
	・向きはない。
・色を読むことで抵抗値を知ることができるが、読み誤りを防ぐためテスタで抵抗値を確認すること

	LED 
	[image: ]
	[image: ]
	・足が長い方を電源の＋側、短い方をアースに接続する。
・方向を逆にすると点灯しない。

	スイッチ
	[image: ]
	[image: ]
	・4個の足のうち、2個を使う。距離が短い2箇所を使う。

	コンデンサ
	[image: ]
	[image: ]
	・向きはない。
・表面に104と印刷されているとき pFである

	水晶発振子
	[image: ]
	[image: ]
	・向きはない
・落として衝撃を与えてはいけない

	圧電ブザー
	[image: ]
	[image: ]
	・ブザーには2種類ある。直流電圧をかけると鳴るものと、交流電圧で駆動するもの。
・ここでは後者を利用する。向きはない。



[image: グラフィカル ユーザー インターフェイス, アプリケーション

自動的に生成された説明]練習
 (1) LEDを光らせる回路
[image: ]






[image: グラフィカル ユーザー インターフェイス, アプリケーション

自動的に生成された説明](2) LEDを2個に増やした回路
[image: ]






 Arduinoとブレッドボードを使った製作
　図3の回路をブレッドボード上に組む。例が図1である。組めたら、以下の手順で確認する。

(1)	LED接続確認用プログラムをArduinoに送り込み、LEDが6個点灯することを確認する。
(2)	ブザー接続確認用プログラムをArduinoに送り込み、ブザーが鳴ることを確認する。
(3)	スイッチ確認用プログラムをArduinoに送り込み、スイッチが正しく動作することを確認する。
(4)	ルーレットプログラムをArduinoに送り込み、ルーレットが正しく動作することを確認する。
 ユニバーサル基板に実装
　ユニバーサル基板に実装する。回路図を図6に示す。
[image: ]
[bookmark: _Ref159949068]図6　ユニバーサル基板上に実装するときの回路図
　ブレッドボード上に組む場合に比べて、22 pFのコンデンサ2個と16 MHzの水晶発振子1個が余分に必要である。
　ユニバーサル基板は多数の穴が空いており、自由にレイアウトすることができる。製作例を図2に示した（図2は最初の試作時のものであり、後述するレイアウト図とは異なる箇所がある）。独自の配線レイアウトをしてもよいが、ここではレイアウト図のサンプルを図7に示す。回路図は素子の「どこ」と「どこ」がつながっているかを示す図である。電気は「つながっているか否か」だけが問題であり、それ以外のことは関係ない。従って、回路図とレイアウト図は形状が大幅に異なることがある。このレイアウト図は一例であり、無数のレイアウトが考え得る。
　赤線で示した部分はユニバーサル基板の裏側における配線を示す。スズメッキ銅線か部品の足を利用して結ぶ。電気は「どことどこが繋がっているか」が問題なので、配線の経路がレイアウト図と異なっていても、接続されている箇所が同一であるなら、全く問題はない。このレイアウト図は交差がないように配線しているが、交差が必要になってしまった場合は、ビニールで被覆された導線を使って配線すればよい。
　配線は基板の裏で行う。基板の表面をスズメッキ銅線で配線することは、混乱を招く恐れがあるので、私はお勧めしない。
[image: ]
[bookmark: _Ref159949281]図7　レイアウト図
　はんだ付けが終わったなら、マイコンを差さない状態で以下のチェックを行う。
抵抗値のチェック
　電源端子（K1）とアース端子（I1）の間の抵抗を測定し、0でない（ショートしてない）ことを確認する。
電圧のチェック
　5 Vの電源電圧を供給した状態で以下の場所の電圧をチェックする。

(1)	1, 7, 20, 15番端子は5 V
	8番端子は0 V
(2)	スイッチを押すと15番端子は0 V
動作チェック
　ルーレットを書き込んだATmega328Pを差し込んで動作チェックする。5 Vの電源はArduinoから取る。

・全く動作しない場合は、水晶発振子、コンデンサの接続をチェックする。
・何らかの動作をする場合は、動作状況に応じて原因を推測し、問題がありそうな箇所をテスタでチェックする。

　完成したら、5 Vの電源をスマホ用のACアダプタなどから取るためのケーブルを自作する。
 付録
ルーレットのプログラム
/* 
ルーレットプログラム
2024.2.28   T. Yabu
*/

const int buttonPin = 9; // ボタンが接続されているピン
const int speakerPin = 8; // スピーカーが接続されているピン
const int ledPins[] = {2, 3, 4, 5, 6, 7}; // LEDが接続されているピンの配列
const int numLeds = sizeof(ledPins) / sizeof(ledPins[0]); // LED の個数
int currentLed = 0; // 現在点灯しているLED

unsigned long start_time;     // 1 回目のボタンを押した時刻
unsigned long slow_down_time; // 2 回目のボタンを押し、減速を開始した時刻
unsigned long stop_time;          // ルーレットを停止する時刻
unsigned long last_led_change_time = 0; // 直前に LED を更新した時間
unsigned long chattering_time = 100;    // 1 回目のボタンから 2 回目のボタンまでの最短時間

int interval = 100;    // LED を更新する間隔 [ms]
int rand_time;         // 2 回目の push から停止するまでの時間
int state = 0;         // 0:停止　1:開始　2:減速中
int input = HIGH;      // 入力
int input_old = HIGH;  // 前回の入力

void setup() {
  int seed;
  pinMode(buttonPin, INPUT);
  pinMode(speakerPin, OUTPUT);
  for (int i = 0; i < numLeds; i++) {
    pinMode(ledPins[i], OUTPUT);
  }

  digitalWrite(ledPins[currentLed], HIGH);

  seed = analogRead(0);
  randomSeed(seed); // 乱数の初期化

  Serial.begin(9600);
  Serial.print("seed : ");
  Serial.println(seed);
}

void loop() {

  unsigned long now;

  now = millis();
  input_old = input;
  input = digitalRead(buttonPin);

  if (input != input_old && input == LOW){  // ボタンを押した
    if ( state == 0 ){   // 停止中の時
      state = 1;
      start_time = now;
      interval = 100;
      digitalWrite(ledPins[currentLed], LOW);
      currentLed = (currentLed + 1) % numLeds;
      digitalWrite(ledPins[currentLed], HIGH);
      tone(speakerPin, 1000, 50); // 移動音
      last_led_change_time = now;
    } else if ( state == 1 ){   // ルーレット定常速度で回転中
      if ( now - start_time > chattering_time ){
        state = 2;
        slow_down_time = now;
        rand_time = random(1000, 7000);  // 停止までの時間 [ms]
        stop_time = now + rand_time; 
      }
    }
  }

  if ( state == 1 || state == 2 ){  // ルーレット回転中
    if ( now - last_led_change_time > interval ){
      digitalWrite(ledPins[currentLed], LOW);
      currentLed = (currentLed + 1) % numLeds;
      digitalWrite(ledPins[currentLed], HIGH);
      tone(speakerPin, 1000, 50); // 移動音
      last_led_change_time = now;
      if ( state == 2 ){
        interval = 100 + ( (now - slow_down_time) / 10); // 徐々に間隔を長くする
        if ( now > stop_time){
          tone(speakerPin, 1000, 200); // ゲーム終了時の音
          state = 0;
        }
      }
    }
  }
}
LEDチェック用
const int ledPins[] = {2, 3, 4, 5, 6, 7}; // LEDが接続されているピンの配列

void setup() {
  for (int i = 0; i < 6; i++) {
    pinMode(ledPins[i], OUTPUT);
  }
}

void loop() {
  for(int i = 0; i < 6; i++){
    digitalWrite(ledPins[i], HIGH);
  }
  delay(500);
  for(int i = 0; i < 6; i++){
    digitalWrite(ledPins[i], LOW);
  }
  delay(500);
}
ブザーチェック用
const int speakerPin = 8; // スピーカーが接続されているピン

void setup() {
  pinMode(speakerPin, OUTPUT);
}

void loop() {
  tone(speakerPin, 1000, 1000);
  delay(2000);
}
ボタンチェック用
const int ledPin = 2;
const int inputPin = 9;

void setup() {
  pinMode(inputPin, INPUT);
  pinMode(ledPin, OUTPUT);
}

void loop() {
  int input = digitalRead(inputPin);

  if (input == LOW){
    digitalWrite(ledPin, HIGH);
  } else {
    digitalWrite(ledPin, LOW);
  }
}
以上

